



SAN JOSE, CA, USA
MARCH 4-7, 2024

Four Problems with Policy-Based Constraints and How to Fix Them

Dillan Mills
Synopsys, Inc.

Chip Haldane
The Chip Abides, LLC



Constraints and Policy Class Review

Constraints Review

- Random objects and constraints are the foundational building blocks of constrained random verification
- Embedded fixed constraints are simple but lack flexibility
- In-line constraints are marginally more flexible but their definitions are still fixed within the calling context
- In-line constraints must all be specified within a single call to `randomize()`

Policy Class Review

- Policy classes are a technique for applying constraints in a portable, reusable, and incremental manner
- Leverage an aspect of "global constraints", simultaneously solving constraints across a set of random objects
- Randomizing a class that contains policies also randomizes the policies
- The policies contain a reference back to the container
- Consequently, the policy container is constrained by the policies it contains

Policy Class Example: policy_base

```
1 class policy_base#(type ITEM=uvm_object);  
2     ITEM item;  
3  
4     virtual function void set_item(ITEM item);  
5         this.item = item;  
6     endfunction  
7 endclass
```

Policy Class Example: policy_list

```
1 class policy_list#(type ITEM=uvm_object) extends policy_base#(ITEM);  
2     rand policy_base#(ITEM) policy[$];  
3  
4     function void add(policy_base#(ITEM) pcy);  
5         policy.push_back(pcy);  
6     endfunction  
7  
8     function void set_item(ITEM item);  
9         foreach(policy[i]) policy[i].set_item(item);  
10    endfunction  
11 endclass
```

policy_base and policy_list Summary

- These two base classes provide the core definitions for policies
- `policy_base` implements the hook back to the policy container
- `policy_list` organizes related policies into groups
- The parameterization requires a unique specialization for each policy-enabled container

Policy Class Example: Implementation

```
1 class addr_txn;  
2   rand addr_t addr;  
3   rand policy_base#(addr_txn) policy[$];  
4  
5   function void pre_randomize;  
6     foreach(policy[i])  
7       policy[i].set_item(this);  
8   endfunction  
9 endclass
```

```
1 class addr_policy extends  
2   ↪ policy_base#(addr_txn);  
3   rand addr_t addrs[$];  
4  
5   function void add(addr_t addr);  
6     addrs.push_back(addr);  
7   endfunction  
8  
9   constraint c_addr {  
10     item.addr inside {addrs};  
11   }  
endclass
```

- `addr_txn.addr` will be constrained to one of the values added through `addr_policy`

Policy Class Example: Usage

```
1 class addr_constrained_txn extends addr_txn;
2   function new;
3     addr_policy addr_policy = new;
4     policy_list#(addr_txn) pcy = new;
5     addr_policy.add('h00000000);
6     addr_policy.add('h10000000);
7     pcy.add(addr_policy);
8     this.policy = {pcy};
9   endfunction
10 endclass
```

- Instantiate and randomize like normal with a call to `txn.randomize()`;
- Each value added to the policy list will be used to constrain the `addr` field

Problem #1: Parameterized Policies

Problem #1: Parameterized Policies

- `policy_base` is parameterized to the class it constrains
- Different specializations cannot be grouped and indexed
- An extended class hierarchy with constrainable values in each level requires a unique policy type and policy list for each level
- Each list must be separately traversed and mapped back to the container during `pre_randomize`
- Users have to keep track of the different lists and which signals apply to each list

Parameterized Policies: Example

```

1  class addr_p_txn extends addr_txn;
2      rand bit parity;
3      rand policy_base#(addr_p_txn)
   ↪  p_policy[$];
4
5      function void pre_randomize;
6          foreach(p_policy[i])
7              p_policy[i].set_item(this);
8      endfunction
9  endclass

```

```

1  class addr_c_txn extends addr_p_txn;
2      function new;
3          policy_list#(addr_txn) pcy = new;
4          policy_list#(addr_p_txn) p_pcy = new;
5          pcy.add(/*addr policies*/);
6          p_pcy.add(/*parity policies*/);
7          this.policy = {pcy};
8          this.p_policy = {p_pcy};
9      endfunction
10 endclass

```

- This method will not scale—each additional subclass requires a new policy type and list

Parameterized Policies: Solution

- Replace the parameterized policy base with a non-parameterized base
- Move the parameterization to an extension class that implements the base

```
1 interface class policy;  
2     pure virtual function void set_item(uvm_object item);  
3 endclass  
4  
5 virtual class policy_imp#(type ITEM=uvm_object)  
6     ↪ implements policy;  
7     protected rand ITEM m_item;  
8  
9     virtual function void set_item(uvm_object item);  
10         if (!$cast(m_item, item)) /* cleanup */;  
11     endfunction  
12 endclass  
13  
14 typedef policy policy_queue[$];
```

Policy Interface and Implementation Classes

- Non-parameterized base enables all policies targeting a particular class hierarchy to be stored in a single common `policy_queue`
- Parameterized `policy_imp` implements the base and provides core functionality required by all policies
- No strong typing means all implementing classes must share a common base class— `uvm_object` is a safe choice for UVM testbenches
- `set_item()` must handle the cases where an invalid type is passed in

Policy Definition Updates

- Policy definitions are mostly still the same as when using parameterized classes
- Policy classes should be updated to extend the new `policy_imp` class

```
1 class addr_policy extends policy_imp#(addr_txn);
```

- Underlying constraints should be written as implications to verify `item` is not missing

```
1 constraint c_addr {m_item != null -> m_item.addr inside {addrs};}
```

Policy Implementation and Usage Updates

- The base txn class needs to inherit from `uvm_object` to be type-compatible
- The `policies` list in the base txn is replaced with a `rand policy_queue policies` declaration
- subclasses of the base txn class no longer need their own policy lists or `pre_randomize()` functions
- The constrained txn can push all policies into the shared `policy_queue` in the base txn class

Policy Usage Example

```
1 class addr_txn extends uvm_object;
2     rand policy_queue policies;
3     // ...
4 class addr_c_txn extends addr_p_txn;
5     function new;
6         // ...
7         this.policies.push_back(/*addr_txn policies*/);
8         this.policies.push_back(/*addr_p_txn policies*/);
9     endfunction
10 endclass
```

Problem #2: Definition Location

Problem #2: Definition Location

- “Where should I define my policy classes?”
- Easy enough to stick them in a file close to the class they are constraining
- Better solution: directly embed policy definitions in the class they constrain
 - Eliminates all guesswork about where to define and discover policies
 - Embedded policies gain access to all members of their container class (including protected properties and methods)

Embedded Policy Example

```
1  class addr_txn extends uvm_object;  
2      class POLICIES;  
3          /* policy definitions */  
4      endclass  
5  endclass  
6  
7  class addr_p_txn extends addr_txn;  
8      class POLICIES extends  
9      ↪ addr_txn::POLICIES;  
10         /* additional policy definitions */  
11     endclass  
12 endclass
```

```
1  class addr_c_txn extends addr_p_txn;  
2      function new;  
3          addr_c_txn::POLICIES::addr_policy  
4      ↪ a_pcy = new(/*...*/);  
5          this.policies.push_back(a_pcy);  
6      endfunction  
7  endclass
```

- Wrap the policies in a POLICIES class to optimize organization
- subclass POLICIES extend from parent POLICIES

Optimize Further

- Mark properties `protected` so they can only be manipulated with policies
- Add static functions to instantiate and initialize policies with a single call

```
1  class addr_txn extends uvm_object;
2      protected rand a_t addr;
3      class POLICIES;
4          // ... addr_policy definition
5          static function addr_policy FIXED_ADDR(a_t a);
6              FIXED_ADDR = new(a);
7          endfunction
8      endclass
9  endclass
10
11 class addr_c_txn extends addr_p_txn;
12     function new;
13         this.policies.push_back(
14             addr_c_txn::POLICIES::FIXED_ADDR('hFF00));
15     endfunction
16 endclass
```

Problem #3: Boilerplate Overload

Problem #3: Boilerplate Overload

- Every policy requires a class definition, a constructor, and a constraint (at a minimum)
- Relatively unavoidable for complex policies
- Generic policy types that show up a lot can be simplified with a macro
- Use macros to set up the embedded `POLICIES` class, the policy definition, and a static constructor

```
1  class addr_p_txn extends addr_txn;  
2      `start_extended_policies(addr_p_txn, addr_txn)  
3          `fixed_policy(PARITY_ERR, parity_err, bit)  
4      `end_policies  
5  endclass
```

Policy Macros

- Ideal for properties with simple constraints, such as equality, range, or set membership constraints
- Complex constraints with relationships between multiple properties are harder to turn into a macro
 - They can be left as-is, defined within the `POLICIES` class
 - They can be moved to a separate file and added to the `POLICIES` class through ``include`

Problem #4: Unexpected Policy Reuse Behavior and Optimizing for Lightweight Policies

Problem #4: Unexpected Policy Reuse Behavior and Optimizing for Lightweight Policies

- Observed occasional unexpected behavior when re-randomizing objects with policies
- Policies seemed to “remember” previous randomizations and wouldn’t reapply constraints
- Sidestep the issue—keep policy classes extremely lightweight and adopt a “use once and discard” approach
- Introduce a `copy()` method that returns a fresh policy instance initialized to the same state as the policy that implements it
- Rely on static constructors to generate initialized policies automatically

More Improvements to the Policy Package

More Improvements to the Policy Package

- Lots of nice improvements to the original policy package so far
- Still lacking many features that would be useful in real-world implementations
- Refer to the paper for complete code examples and a more detailed discussion of the following features

Expanding the policy interface class

```
1 interface class policy;  
2  
3     pure virtual function string name();  
4     pure virtual function string type_name();  
5     pure virtual function string description();  
6     pure virtual function bit item_is_compatible(uvm_object item);  
7     pure virtual function void set_item(uvm_object item);  
8     pure virtual function policy copy();  
9  
10 endclass: policy
```

Better type safety checking and reporting in policy_imp methods

```
1  virtual function void set_item(uvm_object item);  
2      if (item == null)  
3          `uvm_error(/* NULL item passed */)   
4      else if ((this.item_is_compatible(item)) && $cast(this.m_item, item)) begin  
5          `uvm_info(/* Policy applied */)   
6          this.m_item.rand_mode(1);  
7      end else begin  
8          `uvm_warning(/* Incompatible type */)   
9          this.m_item = null;  
10         this.m_item.rand_mode(0);  
11     end  
12 endfunction
```

Replacing `policy_list` with `policy_queue`

- The original policy implementation used a dedicated `policy_list` class to hold policies
- With the new implementation, all you need is a typedef queue

```
1 typedef policy policy_queue[$];
```

- A `policy_queue` can hold any policy that implements the `policy` interface
- Default queue methods can be used to aggregate policies
- array literals can be used where a `policy_queue` is expected
- define, initialize, aggregate, and pass policies all in a single line of code!

Standardize policy implementations with the policy_container interface and policy_object mixin

```
1 interface class policy_container;  
2     pure virtual function bit has_policies();  
3  
4     pure virtual function void set_policies(policy_queue policies);  
5     pure virtual function void add_policies(policy_queue policies);  
6     pure virtual function void clear_policies();  
7  
8     pure virtual function policy_queue get_policies();  
9  
10    pure virtual function policy_queue copy_policies();  
11 endclass
```


Using the policy_container interface class and policy_object mixin

```
1 class policy_object #(type BASE=uvvm_object) extends BASE implements policy_container;  
2     protected policy_queue m_policies;  
3     // ... fill out policy_container functions  
4 endclass
```

```
1 // Use policy_object for transactions  
2 class base_txn extends policy_object #(uvvm_sequence_item);  
3  
4 // Use policy_object for sequences  
5 class base_seq #(type REQ=uvvm_sequence_item, RSP=REQ) extends policy_object #(  
6     ↪ uvvm_sequence#(REQ, RSP) );  
7  
8 // Use policy_object for configuration objects  
9 class cfg_object extends policy_object #(uvvm_object);
```

Protecting the policy queue enforces loosely coupled code

- Using `policy_object` along with `policy_container` allows the policy queue to be protected to prevent direct access
- Original implementation required calling `set_item` on each policy during `pre_randomize`
 - Required because `policy_list` was public so callers could add policies without linking to the target class
- Making it private forces callers to set or add policies with the exposed interface methods
- These methods can also check compatibility and call `set_item` immediately
- Removes the reliance on `pre_randomize`

Conclusion

Conclusion

- Improvements to the original policy package provide a robust and efficient implementation of policy-based constraints
- The policy package is now capable of managing constraints across an entire class hierarchy
- The policy definitions are tightly paired with the class they constrain
- Macros reduce the expense of defining common policies while still allowing flexibility for custom policies
- A complete implementation is available in the Appendix of the paper and on GitHub which can be included directly in a project to start using policies immediately



Questions?